Everybody's Got To Be Somewhere

Conor McBride

Mathematically Structured Programming Group
Department of Computer and Information Sciences
University of Strathclyde, Glasgow

conor.mcbride@strath.ac.uk

The key to any nameless representation of syntax is how it indicates the variables we choose to use and thus, implicitly, those we discard. Standard de Bruijn representations delay discarding *maximally* till the *leaves* of terms where one is chosen from the variables in scope at the expense of the rest. Consequently, introducing new but unused variables requires term traversal. This paper introduces a nameless '*co*-de-Bruijn' representation which makes the opposite canonical choice, delaying discarding *minimally*, as near as possible to the *root*. It is literate Agda: dependent types make it a practical joy to express and be driven by strong intrinsic invariants which ensure that scope is aggressively whittled down to just the *support* of each subterm, in which every remaining variable occurs somewhere. The construction is generic, delivering a *universe* of syntaxes with higher-order *meta*variables, for which the appropriate notion of substitution is *hereditary*. The implementation of simultaneous substitution exploits tight scope control to avoid busywork and shift terms without traversal. Surprisingly, it is also intrinsically terminating, by structural recursion alone.

When I was sixteen and too clever by half, I wrote a text editor which cached a plethora of useful but redundant pointers into the buffer, just to shave a handful of nanoseconds off redisplay. Accurately updating these pointers at each keystroke was a challenge which taught me the hard way about the value of simplicity. Now, I am a dependently typed programmer. I do not keep invariants: invariants keep me.

This paper is about scope invariants in nameless representations of syntax. One motivation for such is eliminating redundant name choice to make α-equivalence trivial. Classic de Bruijn syntaxes [8] replace name by number: variable uses count either out to the binding (*indices*), or in from root to binding (*levels*). Uses are found at the leaves of syntax trees, so any operation which modifies the sequence of variables in scope requires traversal. E.g., consider this β-reduction (under λx) in untyped λ-calculus.

$$\begin{array}{rlcl}
\text{name} & \lambda x.\,(\lambda y.\,y\,x\,(\lambda z.\,z\,(y\,z)))\,\underline{(x\,(\lambda v.\,v))} & \rightsquigarrow_\beta & \lambda x.\,\underline{(x\,(\lambda v.\,v))}\,x\,(\lambda z.\,z\,(\underline{(x\,(\lambda v.\,v))}\,z)) \\
\text{index} & \lambda\ .\,(\lambda\ .\,0\,1\,(\lambda\ .\,0\,(1\,0)))\,\underline{(0\,(\lambda\ .\,0))} & \rightsquigarrow_\beta & \lambda\ .\,\underline{(0\,(\lambda\ .\,0))}\,0\,(\lambda\ .\,0\,(\underline{(1\,(\lambda\ .\,0))}\,0)) \\
\text{level} & \lambda\ .\,(\lambda\ .\,1\,0\,(\lambda\ .\,2\,(1\,2)))\,\underline{(0\,(\lambda\ .\,1))} & \rightsquigarrow_\beta & \lambda\ .\,\underline{(0\,(\lambda\ .\,1))}\,0\,(\lambda\ .\,1\,(\underline{(0\,(\lambda\ .\,2))}\,1))
\end{array}$$

Underlining shows the movement of the substituted term. In the *index* representation, the free x must be shifted when it goes under the λz. With *levels*, the free x stays 0, but the bound v must be shifted under λz, and the substitution context must be shifted to account for the eliminated λy. Shift happens.

The objective of this paper is not to eliminate shifts altogether, but to ensure that they do not require traversal. The approach is to track exactly which variables are *relevant* at all nodes in the tree and aggressively expel those unused in any given subtree. As we do so, we need and obtain much richer accountancy of variable usage, with much more intricate invariants. Category theory guides the design of these invariants and Agda's dependent types [18] drive their correct implementation.

My explorations follow Sato, Pollack, Schwichtenberg and Sakurai, whose λ-terms make binding sites carry *maps* of use sites [19]. E.g., the $\mathbb{K}$ and $\mathbb{S}$ combinators become (respectively)

names	$\lambda c.\,\lambda e.\,c$	$\lambda f.$	$\lambda s.$	$\lambda e.\ (f\,e)\ \ (s\,e)$
maps	`1\ 0\□`	`((10) (00))\((00) (10))\((01) (01))\`		$(\square\,\square)\,(\square\,\square)$

Submitted to:
MSFP 2018

where each abstraction shows with `1`s where in the subsequent tree of applications its variable occurs: leaves, $\Box$, are relieved of choice. Of course, the tree under each binder determines which maps are well formed in a highly nonlocal way: these invariants are formalised *extrinsically* both in Isabelle/HOL and in Minlog, over a context-free datatype enforcing neither scope nor shape.

In this paper, we shall obtain an *intrinsically* valid representation, where the map information is localized. Binding sites tell only if the variable is used; the crucial choice points where a term comprises more than one subterm say which variables go where. Not all are used in all subterms, but (as Eccles says to Seagoon) *everybody's got to be somewhere* [17]: variables used nowhere have been discarded already. This property is delivered by a coproduct construction in the slices of the category of order-preserving embeddings, but fear not: we shall revisit all of the category theory required to develop the definition, especially as it strays beyond the familiar (e.g., to Haskellers) territory of types-and-functions.

Intrinsically well scoped de Bruijn terms date back to Bellegarde and Hook [5], using `option` types to grow a type of free variables, but hampered by lack of polymorphic recursion in ML. Substitution (i.e., *monadic* structure) was developed for untyped terms by Bird and Paterson [7] and for simple types by Altenkirch and Reus [4], both dependent either on a *prior* implementation of renumbering shifts (i.e., functorial structure) or a non-structural recursion. My thesis [15] follows McKinna and Goguen [11] in restoring a single structural operation abstracting 'action' on variables, instantiated to renumbering then to substitution, an approach subsequently adopted by Benton, Kennedy and Hur [6] and generalised to semantic actions by Allais et al. [2]. Here, we go directly to substitution: *shifts need no traversal*.

I present not only λ-calculus but a *universe* of syntaxes inspired by Harper, Honsell and Plotkin's Logical Framework [12]. I lift the *sorts* of a syntax to higher *kinds*, acquiring both binding (via subterms at higher kind) and *meta*variables (at higher kind). However, substituting a higher-kinded variable demands substitution of its parameters *hereditarily* [21] and *simultaneously*. Thereby hangs a tale. Abel showed how *sized types* justify this process's apparently non-structural recursion in MSFP 2006 [1]. As editor, I anonymised a discussion with a referee which yielded a structural recursion for hereditary substitution of a *single* variable, instigating Keller and Altenkirch's formalization at MSFP 2010 [14]. Here, at last, simultaneous hereditary substitution becomes structurally recursive.

1 Basic Equipment in Agda

We shall need finite types Zero, One, and Two, named for their cardinality, and the reflection of Two as a set of evidence for 'being tt'. Dependent pairing is by means of the Σ type, abbreviated by $\times$ when non-dependent. The *pattern synonym* !_ allows the first component to be determined by the second: making it a right-associative prefix operator lets us write ! ! *expression* rather than ! (! (*expression*)).

```
data  Zero : Set where
record One : Set where constructor ⟨⟩
data  Two : Set where tt ff : Two

Tt : Two → Set
Tt tt = One
Tt ff = Zero
```

```
record Σ (S : Set) (T : S → Set) : Set where
  constructor _,_
  field fst : S; snd : T fst

_×_ : Set → Set → Set
S × T = Σ S λ _ → T
pattern !_ t = _ , t
```

We shall also need to reason equationally. For all its imperfections in matters of *extensionality*, it will be convenient to define equality inductively, enabling the **rewrite** construct in equational proofs.

```
data _==_ {X : Set} (x : X) : X → Set where refl : x == x
```

2 Δ_+^K: The (Semi-Simplicial) Category of Order-Preserving Embeddings

No category theorist would mistake me for one of their own. However, the key technology in this paper can be helpfully conceptualised categorically. Category theory is just the study of compositionality — for everything, not just sets-and-functions. Here, we have an opportunity to develop categorical structure away from the usual apparatus for programming with functions. Let us therefore revisit the basics.

Category (I): Objects and Morphisms. A *category* is given by a class of *objects* and a family of *morphisms* (or *arrows*) indexed by two objects: *source* and *target*. Abstractly, we may write $\mathbb{C}$ for a given category, $|\mathbb{C}|$ for its objects, and $\mathbb{C}(S,T)$ for its morphisms with given source and target, $S,T \in |\mathbb{C}|$.

The rest will follow, but let us fix these notions for our example category, Δ_+^K, of *order-preserving embeddings* between variable scopes, which I learned about from Altenkirch, Hofmann and Streicher [3]. Objects are *scopes*, given as backward (or 'snoc') lists of the *kinds*, K, of variables. (I habitually suppress the K and just write Δ_+ for the category.) Backward lists respect the tradition of writing contexts left of judgements in rules and extending them on the right. However, I write 'scope' rather than 'context' as we track at least which variables we may refer to, but perhaps not all contextual data.

```
data Bwd (K : Set) : Set where
  _-,_ : Bwd K → K → Bwd K
  []   : Bwd K
```

```
data _≤_ : Bwd K → Bwd K → Set where
  _o'  : iz ≤ jz →  iz      ≤ (jz -, k)
  _os  : iz ≤ jz → (iz -, k) ≤ (jz -, k)
  oz   :                []  ≤  []
```

The morphisms, $iz \le jz$, of Δ_+ give an embedding from a source to a target scope. Colloquially, we may call them 'thinnings', as they dilute the variables of the source scope with more. Dually, we may see such a morphism as expelling variables from the target scope, leaving a particular selection as the source. I write the step constructors postfix, so thinnings (like scopes) grow on the right. When $K = \mathsf{One}$, $\mathsf{Bwd}\ K$ represents numbers and $\le$ generates Pascal's Triangle; excluding the empty scope and allowing *degenerate* (non-injective) maps yields Δ, the *simplex* category beloved of topologists.

Now, where I give myself away as a type theorist is that I do not consider the notion of 'morphism' to make sense without prior source and target objects. The type $iz \le jz$ (which is a little more mnemonic than $\Delta_+(iz,jz)$) is the type of 'thinnings from iz to jz': there is no type of 'thinnings' *per se*.

Let us have an example thinning: here, we embed a scope with three variables into a scope with five.



Category (II): Identity and Composition. In any category, certain morphisms must exist. Each object $X \in |\mathbb{C}|$ has an *identity* $\iota_X \in \mathbb{C}(X,X)$, and wherever the target of one morphism meets the source of another, their *composite* makes a direct path: if $f \in \mathbb{C}(R,S)$ and $g \in \mathbb{C}(S,T)$, then $(f;g) \in \mathbb{C}(R,T)$.

For example, every scope has the identity thinning, oi, and thinnings compose via ⨟. (For functions, it is usual to write $g \cdot f$ for 'g *after* f' than $f;g$ for 'f *then* g', but for thinnings I retain spatial intuition.)

```
oi : kz ≤ kz
oi {kz = iz -, k} = oi os   -- os preserves oi
oi {kz = []}      = oz
```

```
_⨟_ : iz ≤ jz → jz ≤ kz → iz ≤ kz
θ    ⨟ ϕ o'  = (θ ⨟ ϕ) o'
θ o' ⨟ ϕ os  = (θ ⨟ ϕ) o'
θ os ⨟ ϕ os  = (θ ⨟ ϕ) os   -- os preserves ⨟
oz   ⨟ oz    = oz
```

By way of example, let us plot specific uses of identity and composition.



Category (III): Laws. To complete the definition of a category, we must say which laws are satisfied by identity and composition. Composition *absorbs* identity on the left and on the right. Moreover, composition is *associative*, meaning that any sequence of morphisms which fit together target-to-source can be composed without the specific pairwise grouping choices making a difference. That is, we have three laws which are presented as *equations*, at which point any type theorist will want to know what is meant by 'equal': I shall always be careful to say. Our thinnings are first-order, so $\equiv$ will serve. With this definition in place, we may then state the laws. I omit the proofs, which go by functional induction.

```
law−oi⨟ : oi ⨟ θ ≡ θ      law−⨟oi : θ ⨟ oi ≡ θ      law−⨟⨟ : θ ⨟ (ϕ ⨟ ψ) ≡ (θ ⨟ ϕ) ⨟ ψ
```

As one might expect, order-preserving embeddings have a strong antisymmetry property that one cannot expect of categories in general. The *only* invertible arrows are the identities. Note that we must match on the proof of $iz \equiv jz$ even to claim that θ and ϕ are the identity.

```
antisym : (θ : iz ≤ jz) (ϕ : jz ≤ iz) → Σ (iz ≡ jz) λ {refl → θ ≡ oi × ϕ ≡ oi}
```

Example: de Bruijn Syntax via Δ_+^{One}. De Bruijn indices are numbers [8], perhaps with some bound enforced by type [5, 7, 4]. We can use singleton thinning, $k \leftarrow kz = [] \text{-}, k \leq kz$, to give de Bruijn λ-terms, readily admitting thinning:

```
data Lam (iz : Bwd One) : Set where
  #   : (x : ⟨⟩ ← iz) →       Lam iz
  _$_ : (f s : Lam iz) →      Lam iz
  λ   : (t : Lam (iz -, ⟨⟩)) → Lam iz
```

```
_↑_ : Lam iz → iz ≤ jz → Lam jz
# i     ↑ θ  = # (i ⨟ θ)
(f $ s) ↑ θ  = (f ↑ θ) $ (s ↑ θ)
λ t     ↑ θ  = λ (t ↑ θ os)
```

Variables are represented by pointing, eliminating redundant choice of names, but it is only when we point to one variable that we exclude the others. Thus de Bruijn indexing effectively uses thinnings to discard unwanted variables as *late* as possible, in the *leaves* of syntax trees.

Note how the scope index iz is the target of a thinning in # and weakened in λ. Hence, thinnings act on terms ultimately by postcomposition, but because terms keep their thinnings at their leaves, we must hunt the entire tree to find them. Now consider the other canonical placement of thinnings, nearest the *root*, discarding unused variables as *early* as possible.

3 Slices of Thinnings

If we fix the target of thinnings, $(\_ \leq kz)$, we obtain the notion of *subscopes* of a given kz. Fixing a target is a standard way to construct a new category whose objects are given by morphisms of the original.

Slice Category. If $\mathbb{C}$ is a category and I one of its objects, the *slice category* $\mathbb{C}/I$ has as its objects pairs (S, f), where S is an object of $\mathbb{C}$ and $f : S \to I$ is a morphism in $\mathbb{C}$. A morphism in $(S, f) \to (T, g)$ is some $h : S \to T$ such that $f = h; g$. (The dotted regions in the diagram show the objects in the slice.)



That is, the morphisms are *triangles*. A seasoned dependently typed programmer will be nervous at a definition like the following (where the _ after Σ asks Agda to compute the type $iz \leq jz$ of θ):

```
ψ →/ φ = Σ _ λ θ → (θ ⨟ φ) == ψ   -- beware of ⨟!
```

because the equation restricts us when it comes to manipulating triangles. Dependent pattern matching relies on *unification* of indices, but defined function symbols like ⨟ make unification difficult, obliging us to reason about the *edges* of the triangles. It helps at this point to define the *graph* of ⨟ inductively.

```
data Tri : iz ≤ jz → jz ≤ kz → iz ≤ kz → Set where
  _t-"  : Tri θ φ ψ → Tri  θ     (φ o') (ψ o')
  _t's' : Tri θ φ ψ → Tri (θ o') (φ os) (ψ o')
  _tsss : Tri θ φ ψ → Tri (θ os) (φ os) (ψ os)
  tzzz  :             Tri  oz     oz      oz

tri  : (θ : iz ≤ jz) (φ : jz ≤ kz) →
       Tri θ φ (θ ⨟ φ)
comp : Tri θ φ ψ → ψ == (θ ⨟ φ)
```

The indexing is entirely in constructor form, which will allow easy unification. Moreover, all the *data* in a Tri structure comes from its *indices*. Easy inductions show that Tri is precisely the graph of ⨟.

The example composition given above can be rendered a triangle, as follows:

```
egTri : Tri {kz = [] -, k0 -, k1 -, k2 -, k3 -, k4} (oz os o' os) (oz os o' os o' os) (oz os o' o' o' os)
egTri = tzzz tsss t-" t's' t-" tsss
```

Morphisms in the slice can now be triangles: $\psi \to_/ \phi = \Sigma \_ \lambda\ \theta \to \mathsf{Tri}\ \theta\ \phi\ \psi$.

A useful Δ_+-specific property is that morphisms in Δ_+/kz are *unique*. It is easy to state this property in terms of triangles with common edges, $\mathsf{triU} : \mathsf{Tri}\ \theta\ \phi\ \psi \to \mathsf{Tri}\ \theta'\ \phi\ \psi \to \theta == \theta'$, and then prove it by induction on the triangles, not edges. It is thus cheap to obtain *universal properties* in the slices of Δ_+, asserting the existence of unique morphisms: uniqueness comes for free!

4 A Proliferation of Functors

Haskell makes merry with `class Functor` and its many subclasses: this scratches but the surface, giving only *endo*functors from types-and-functions to types-and-functions. Once we adopt the general notion, functoriality sprouts everywhere, with the same structures usefully functorial in many ways.

Functor. A *functor* is a mapping from a source category $\mathbb{C}$ to a target category $\mathbb{D}$ which preserves categorical structure. To specify a structure, we must give a function $F_o : |\mathbb{C}| \to |\mathbb{D}|$ from source objects to target objects, together with a family of functions $F_m : \mathbb{C}(S, T) \to \mathbb{D}(F_o(S), F_o(T))$. The preserved

structure amounts to identity and composition: we must have that $F_m(\iota_X) = \iota_{F_o(X)}$ and that $F_m(f;g) = F_m(f);F_m(g)$. Note that there is an identity functor $\mathbf{I}$ (whose actions on objects and morphisms are the identity) from $\mathbb{C}$ to itself and that functors compose (componentwise).

E.g., every $k : K$ induces a functor (*weakening*) from Δ_+ to itself by scope extension, $(\_\ \text{-},\ k)$ on objects and os on morphisms. The very definitions of oi and ⨟ show that os preserves oi and ⨟.

To see more examples, we need more categories. Let Set's objects be types in Agda's Set universe and $\mathsf{Set}(S,T)$ exactly $S \to T$, with the usual identity and composition. Morphism equality is *pointwise*. Exercises: make $\mathsf{Bwd} : \mathsf{Set} \to \mathsf{Set}$ a functor; check $(\mathsf{Lam}, \uparrow)$ is a functor from Δ_+ to Set.

Let us plough a different furrow, rich in dependent types, constructing new categories by *indexing*. If $I : \mathsf{Set}$, we may then take $I \to \mathsf{Set}$ to be the category whose objects are *families* of objects in Set, $S, T : I \to \mathsf{Set}$ with morphisms (implicitly indexed) families of functions: $S \dot{\to} T = \forall\,\{i\} \to S\,i \to T\,i$. Morphisms are equal if they map each index to pointwise equal functions.

We may define a functor from $K \to \mathsf{Set}$ to $\mathsf{Bwd}\,K \to \mathsf{Set}$ as follows:

```
data All (P : K → Set) : Bwd K → Set where
  []   :                    All P []
  _-,_ : All P kz → P k → All P (kz -, k)
```

```
all : (P ⇾ Q) → (All P ⇾ All Q)
all f []        = []
all f (pz -, p) = all f pz -, f p
```

For a given K, All acts on objects, giving for each k in a scope, some value in $P\,k$, thus giving us a notion of *environment*. The action on morphisms, all, lifts *kind-respecting* operations on values to *scope-respecting* operations on environments. Identity and composition are readily preserved. In the sequel, it will be convenient to abbreviate $\mathsf{Bwd}\,K \to \mathsf{Set}$ as $\overline{K}$, for types indexed over scopes.

However, All gives more functorial structure. Fixing kz, we obtain $\lambda\ P \to \mathsf{All}\ P\ kz$, a functor from $K \to \mathsf{Set}$ to Set, again with the instantiated all acting on morphisms. And still, there is more.

Opposite Category. For a given category $\mathbb{C}$, its *opposite* category is denoted $\mathbb{C}^{\mathrm{op}}$ and defined thus:

$$|\mathbb{C}^{\mathrm{op}}| = |\mathbb{C}| \qquad \mathbb{C}^{\mathrm{op}}(S,T) = \mathbb{C}(T,S) \qquad \iota_X^{\mathrm{op}} = \iota_X \qquad f;^{\mathrm{op}} g = g;f$$

Note that a functor from $\mathbb{C}^{\mathrm{op}}$ to $\mathbb{D}$ is sometimes called a *contravariant functor* from $\mathbb{C}$ to $\mathbb{D}$.

E.g., $\Delta_+^{\mathrm{op}}(jz, iz) = iz \le jz$ views thinnings as *selections* of just iz from the jz on offer. As shown, right, an environment for all the jz whittles down to just the iz, making $\mathsf{All}\,P$ a *presheaf* on Δ_+ — a *functor* from Δ_+^{op} to Set.

```
_≤?_ : iz ≤ jz → All P jz → All P iz
oz     ≤? []        = []
(θ os) ≤? (pz -, p) = (θ ≤? pz) -, p
(θ o') ≤? (pz -, p) = θ ≤? pz
```

Natural Transformation. Given functors F and G from $\mathbb{C}$ to $\mathbb{D}$, a *natural transformation* is a family of $\mathbb{D}$-morphisms $k_X \in \mathbb{D}(F_o(X), G_o(X))$ indexed by $\mathbb{C}$-objects, $X \in |\mathbb{C}|$, satisfying the *naturality* condition, which is that for any $h \in \mathbb{C}(S,T)$, we have $k_S; G_m(h) = F_m(h); k_T$, amounting to a kind of *uniformity*. It tells us that k_X does not care what X is, but only about the additional structure imposed by F and G.

Parametric polymorphism famously induces naturality [20], in that ignorance of a parameter imposes uniformity, and the same is true in our more nuanced setting. We noted that $\lambda\ P \to \mathsf{All}\ P\ kz$ is a functor (with action all) from $K \to \mathsf{Set}$ to Set. Accordingly, if $\theta : iz \le jz$ then $(\theta \le ?_)$is a natural transformation from $\lambda\ P \to \mathsf{All}\ P\ jz$ to $\lambda\ P \to \mathsf{All}\ P\ iz$, which is as much as to say that the definition of $\le ?$ is uniform in P, and hence that if $f : \forall\,\{k\} \to P\,k \to Q\,k$, then $\mathsf{all}\,f\ (\theta \le ?\ pz) = \theta \le ?\ \mathsf{all}\,f\ pz$.

Dependently typed programming thus offers us a richer seam of categorical structure than we see in Haskell. This presents an opportunity to make sense of the categorical taxonomy in terms of concrete programming examples, and at the same time, organising those programs and indicating *what to prove*.

5 Things-with-Thinnings (a Monad)

Let us develop the habit of packing terms with an object in the slice category of thinnings, selecting the support of the term and discarding unused variables at the root. Note that $\Uparrow$ is a functor from $\overline{K}$ to itself.

```
record _⇑_ {K} (T : K̄) (scope : Bwd K) : Set where    -- (T ⇑_) : K̄
  constructor _↑_
  field {support} : Bwd K;   thing : T support;   thinning : support ≤ scope
map⇑ : ∀ {K} {S T : K̄} → (S →̇ T) → ((S ⇑_) →̇ (T ⇑_))
map⇑ f (s ↑ θ) = f s ↑ θ
```

In fact, the categorical structure of Δ_+ makes $\Uparrow$ a *monad*. Let us recall the definition.

Monad. A functor M from $\mathbb{C}$ to $\mathbb{C}$ gives rise to a *monad* (M, η, μ) if we can find a pair of natural transformations, respectively 'unit' ('add an M layer') and 'multiplication' ('merge M layers').

$$\eta_X : \mathbf{I}(X) \to M(X) \qquad \mu_X : M(M(X)) \to M(X)$$

subject to the conditions that merging an added layer yields the identity (whether the layer added is 'outer' or 'inner'), and that adjacent M layers may be merged pairwise in any order.

$$\eta_{M(X)}; \mu_X = \iota_{M(X)} \qquad M(\eta_X); \mu_X = \iota_{M(X)} \qquad \mu_{M(X)}; \mu_X = M(\mu_X); \mu_X$$

The categorical structure of thinnings makes $\Uparrow$ a monad. Here, 'adding a layer' amounts to 'wrapping with a thinning'. The proof obligations to make $(\Uparrow, \mathsf{unit}\Uparrow, \mathsf{mult}\Uparrow)$ a monad are exactly those required to make Δ_+ a category in the first place. In particular, things-with-thinnings are easy to thin further, indeed, parametrically so. In other words, $(T \Uparrow)$ is uniformly a functor from Δ_+ to Set.

```
unit⇑ : T →̇ (T ⇑_)      mult⇑ : ((T ⇑_) ⇑_) →̇ (T ⇑_)      thin⇑ : iz ≤ jz → T ⇑ iz → T ⇑ jz
unit⇑ t = t ↑ oi         mult⇑ ((t ↑ θ) ↑ ϕ) = t ↑ (θ ;; ϕ)   thin⇑ θ t = mult⇑ (t ↑ θ)
```

Shortly, we shall learn how to find the variables on which a term syntactically depends. However, merely *allowing* a thinning at the root, Lam $\Uparrow$ *iz*, yields a redundant representation, as we may discard variables at either root or leaves. Let us eliminate redundancy by *insisting* that a term's support is *relevant*: a variable retained by the thinning *must* be used in the thing. Everybody's got to be somewhere.

6 The Curious Case of the Coproduct in Slices of Δ_+

The $\Uparrow$ construction makes crucial use of objects in the slice category $\Delta_+/scope$, which exhibit useful additional structure: they are *bit vectors*, with one bit per variable telling whether it has been selected. Bit vectors inherit Boolean structure, via the 'Naperian' array structure of vectors [10].

Initial object. A category $\mathbb{C}$ has initial object 0, if there is a unique morphism in $\mathbb{C}(0, X)$ for every X.

The *empty type* is famed for this rôle for types-and-functions: empty case analysis gives the vacuously unique morphism. In Δ_+, the empty *scope* plays this rôle, with the 'constant 0' bit vector as unique morphism. By return of post, we get $([], \mathsf{oe})$ as the initial object in the slice category Δ_+/kz. Hence, we

can make *constants* with empty support, i.e., noting that no variable is ($\cdot_R$ for) *relevant*.

```
oe : ∀ {K} {kz : Bwd K} → [] ≤ kz          oe/ : (θ : iz ≤ kz) → oe →/ θ
oe {kz = iz -, k} = oe o'                  oe/ θ with tri oe θ
oe {kz = []}      = oz                     ... | t rewrite law−oe (oe ; θ) = oe , t

law−oe : (θ : [] ≤ kz) → θ == oe

data One_R {K} : K̄ where ⟨⟩ : One_R []     ⟨⟩_R : One_R ⇑ kz;   ⟨⟩_R = ⟨⟩ ↑ oe
```

We should expect the constant to be the trivial case of some notion of *relevant pairing*, induced by *coproducts* in the slice category. If we have two objects in Δ_+/kz representing two subscopes, there should be a smallest subscope which includes both: pairwise disjunction of bit vectors.

Coproduct. Objects S and T of category $\mathbb{C}$ have a coproduct object $S+T$ if there are morphisms $l \in \mathbb{C}(S, S+T)$ and $r \in \mathbb{C}(T, S+T)$ such that every pair $f \in \mathbb{C}(S,U)$ and $g \in \mathbb{C}(T,U)$ factors through a unique $h \in \mathbb{C}(S+T,U)$ so that $f = l;h$ and $g = r;h$. In Set, we may take $S+T$ to be the *disjoint union* of S and T, with l and r its injections and h the *case analysis* whose branches are f and g.

However, we are not working in Set, but in a slice category. Any category theorist will tell you that slice categories $\mathbb{C}/I$ inherit *colimit* structure (characterized by universal out-arrows) from $\mathbb{C}$, as indeed we just saw with the initial object. If Δ_+ has coproducts, too, we are done!

Curiously, though, Δ_+ does *not* have coproducts. Taking $K = \mathsf{One}$, let us seek the coproduct of two singletons, $S = T = [] \text{-}, \langle\rangle$. Construct one diagram by taking $U = [] \text{-}, \langle\rangle$ and $f = g = \mathsf{oi}$, ensuring that our only candidate for $S+T$ is again the singleton $[] \text{-}, \langle\rangle$, with $l = r = \mathsf{oi}$, making $h = \mathsf{oi}$. Nothing else can sit between S, T and U. Now begin a different diagram, with $U' = [] \text{-}, \langle\rangle \text{-}, \langle\rangle$, allowing $f' = \mathsf{oz\ os\ o'}$ and $g' = \mathsf{oz\ o'\ os}$. No h' post-composes l and r (both oi, making h' itself) to yield f' and g' respectively.

Fortunately, we get what we need: Δ_+ may not have coproducts, but its *slices* do. Examine the data: two subscopes of some kz, $\theta : iz \le kz$ and $\phi : jz \le kz$. Their coproduct must be some $\psi : ijz \le kz$, where our l and r must be triangles $\mathsf{Tri}\ \theta'\ \psi\ \theta$ and $\mathsf{Tri}\ \phi'\ \psi\ \phi$, giving morphisms in $\theta \to_/ \psi$ and $\phi \to_/ \psi$. Choose ψ to be pointwise disjunction of θ and ϕ, minimizing ijz: θ' and ϕ' will then *cover* ijz.

```
data Cover {K} (ov : Two) : {iz jz ijz : Bwd K} → iz ≤ ijz → jz ≤ ijz → Set where
  _c's  :                    Cover ov θ φ → Cover ov (θ o') (φ os)
  _cs'  :                    Cover ov θ φ → Cover ov (θ os) (φ o')
  _css  : {both : Tt ov} →   Cover ov θ φ → Cover ov (θ os) (φ os)
  czz   :                                  Cover ov oz oz
```

The flag, ov, determines whether *overlap* is permitted:tt for coproducts. Cover ff specifies *partitions*. No constructor allows both θ and ϕ to omit a target variable, so everybody's got to be somewhere. Let us compute the coproduct, ψ then check that any other diagram for some ψ' yields a $\psi \to_/ \psi'$.

```
cop  : (θ : iz ≤ kz) (φ : jz ≤ kz) →
       Σ _ λ ijz → Σ (ijz ≤ kz) λ ψ →
       Σ (iz ≤ ijz) λ θ' → Σ (jz ≤ ijz) λ φ' →
       Tri θ' ψ θ × Cover tt θ' φ' × Tri φ' ψ φ
copU : Tri θ' ψ θ → Cover tt θ' φ' → Tri φ' ψ φ →
       θ →/ ψ' → φ →/ ψ' → ψ →/ ψ'
```



```
cop (θ o') (ϕ o')  = let ! ! ! ! tl , c , tr = cop θ ϕ in ! ! ! ! tl t-"  , c     , tr t-"
cop (θ o') (ϕ os)  = let ! ! ! ! tl , c , tr = cop θ ϕ in ! ! ! ! tl t's' , c c's , tr tsss
cop (θ os) (ϕ o')  = let ! ! ! ! tl , c , tr = cop θ ϕ in ! ! ! ! tl tsss , c cs' , tr t's'
cop (θ os) (ϕ os)  = let ! ! ! ! tl , c , tr = cop θ ϕ in ! ! ! ! tl tsss , c css , tr tsss
cop    oz     oz   =                                      ! ! ! !   tzzz ,   czz ,   tzzz
```

The copU proof goes by induction on the triangles which share ψ' and inversion of the coproduct. The payoff from the coproduct construction is the type of *relevant pairs* — the co-de-Bruijn touchstone:

```
record _×R_ (S T : K̄) (ijz : Bwd K) : Set where
  constructor pair
  field outl  : S ⇑ ijz;   outr : T ⇑ ijz
        cover : Cover tt (thinning outl) (thinning outr)

_,R_ : S ⇑ kz → T ⇑ kz → (S ×R T) ⇑ kz
(s ↑ θ) ,R (t ↑ ϕ) =
  let ! ψ , θ' , ϕ' , _ , c , _ = cop θ ϕ
  in  pair (s ↑ θ') (t ↑ ϕ') c ↑ ψ
```

7 Monoidal Structure of Order-Preserving Embeddings

To talk about binding, we need scope extension. We have seen single 'snoc', but binding is simultaneous in general. Concatenation induces monoidal structure on objects of Δ_+, extending to morphisms.

```
_++_ : (iz jz : Bwd K) → Bwd K
kz ++ []      = kz
kz ++ (iz -, j) = (kz ++ iz) -, j

_++≤_ : iz ≤ jz → iz' ≤ jz' → (iz ++ iz') ≤ (jz ++ jz')
θ ++≤ oz     = θ
θ ++≤ (ϕ os) = (θ ++≤ ϕ) os
θ ++≤ (ϕ o') = (θ ++≤ ϕ) o'
```

Moreover, given a embedding into a concatenation, we can split it into local and global parts.

```
_⊣_ : (jz : Bwd K) (ψ : iz ≤ (kz ++ jz)) → Σ (Bwd K) λ kz' → Σ (Bwd K) λ jz' →
  Σ (kz' ≤ kz) λ θ → Σ (jz' ≤ jz) λ ϕ → Σ (iz == (kz' ++ jz')) λ {refl → ψ == (θ ++≤ ϕ)}
[]      ⊣ ψ                                  = ! ! ψ , oz , refl , refl
(kz -, k) ⊣ (ψ os)          with kz ⊣ ψ
(kz -, k) ⊣ (. (θ ++≤ ϕ) os) | ! ! θ , ϕ , refl , refl = ! ! θ , ϕ os , refl , refl
(kz -, k) ⊣ (ψ o')          with kz ⊣ ψ
(kz -, k) ⊣ (. (θ ++≤ ϕ) o') | ! ! θ , ϕ , refl , refl = ! ! θ , ϕ o' , refl , refl
```

Thus equipped, we can say how to bind some variables. The key is to say at the binding site which of the bound variables will actually be used: if they are not used, we should not even bring them into scope.

```
data _⊢_ {K} (jz : Bwd K) (T : K̄) (kz : Bwd K) : Set where    -- jz ⊢ T : K̄
  _\\_ : ∀ {iz} → iz ≤ jz → T (kz ++ iz) → (jz ⊢ T) kz

_\\R_ : ∀ {K T} {kz} (jz : Bwd K) → T ⇑ (kz ++ jz) → (jz ⊢ T) ⇑ kz
jz \\R (t ↑ ψ) with jz ⊣ ψ
jz \\R (t ↑ . (θ ++≤ ϕ)) | ! ! θ , ϕ , refl , refl = (ϕ \\ t) ↑ θ
```

The monoid of scopes is generated from its singletons. By the time we *use* a variable, it should be the only thing in scope. The associated smart constructor computes the thinned representation of variables.

```
data VaR (k : K) : K̄ where
  only : VaR k ([] -, k)

vaR : k ← kz → VaR k ⇑ kz
vaR x = only ↑ x
```

Untyped λ-calculus. We can now give the λ-terms for which all *free* variables are relevant as follows. Converting de Bruijn to co-de-Bruijn representation is easy with smart constructors. E.g., compare de Bruijn terms for the $\mathbb{K}$ and $\mathbb{S}$ combinators with their co-de-Bruijn form.

```
data Lam_R : One where
  #   : Va_R ⟨⟩              → Lam_R
  app : (Lam_R ×_R Lam_R)    → Lam_R
  λ   : ([] -, ⟨⟩ ⊢ Lam_R)   → Lam_R

lam_R : Lam ⇢ (Lam_R ⇑ _)
lam_R (# x)   = map⇑ #   (va_R x)
lam_R (f $ s) = map⇑ app (lam_R f ,_R lam_R s)
lam_R (λ t)   = map⇑ λ   (_ \\_R lam_R t)

𝕂       = λ (λ (# (oe os o')))
lam_R 𝕂 = λ (oz os \\ λ (oz o' \\ # only)) ↑ oz
𝕊       = λ (λ (λ (# (oe os o' o') $ # (oe os) $ (# (oe os o') $ # (oe os)))))
lam_R 𝕊 = λ (oz os \\ λ (oz os \\ λ (oz os \\
  app (pair (app (pair (# only ↑ oz os o') (# only ↑ oz o' os) (czz cs' c's)) ↑ oz os o' os)
            (app (pair (# only ↑ oz os o') (# only ↑ oz o' os) (czz cs' c's)) ↑ oz o' os os)
            (czz cs' c's css))))) ↑ oz
```

Stare bravely! $\mathbb{K}$ returns a plainly constant function. Meanwhile, $\mathbb{S}$ clearly uses all three inputs: the function goes left, the argument goes right, and the environment is shared.

8 A Universe of Metasyntaxes-with-Binding

There is nothing specific to the λ-calculus about de Bruijn representation or its co-de-Bruijn counterpart. We may develop the notions generically for multisorted syntaxes. If the sorts of our syntax are drawn from set I, then we may characterize terms-with-binding as inhabiting Kinds $kz \Rightarrow i$, which specify an extension of the scope with new bindings kz and the sort i for the body of the binder.

```
record Kind (I : Set) : Set where inductive;  constructor _⇒_
                                    field scope : Bwd (Kind I);  sort : I
```

Kinds offer higher-order abstraction: a bound variable itself has a Kind, being an object sort parametrized by a scope, where the latter is, as in previous sections, a Bwd list, with K now fixed as Kind I. Object variables have sorts; *meta*-variables have Kinds. E.g., in the β-rule, t and s are not object variables like x

$$(\lambda x.\, t[x])\, s \rightsquigarrow t[s]$$

but placeholders, s for some term and $t[x]$ for some term with a parameter which can be and is instantiated, by x on the left and s on the right. The kind of t is [] -, ([] ⇒ ⟨⟩) ⇒ ⟨⟩.

We may give the syntax of each sort as a function mapping sorts to Descriptions $D : I \to$ Desc I.

```
data Desc (I : Set) : Set_1 where
  Rec_D : Kind I → Desc I;   Σ_D   : (S : Datoid) → (Data S → Desc I) → Desc I
  One_D :          Desc I;   _×_D_ : Desc I → Desc I →                 Desc I
```

We may ask for a subterm with a given Kind, so it can bind variables by listing their Kinds left of ⇒. Descriptions are closed under unit and pairing. We may also ask for terms to be tagged by some sort of 'constructor' inhabiting some Datoid, i.e., a set with a decidable equality, given as follows:

```
data Decide (X : Set) : Set where
  yes : X →          Decide X
  no  : (X → Zero) → Decide X

record Datoid : Set_1 where
  field Data   : Set
        decide : (x y : Data) → Decide (x == y)
```

Describing untyped λ-calculus. Define a tag enumeration, then a description.

```
data LamTag : Set where app λ : LamTag

LAMTAG : Datoid
Data LAMTAG = LamTag

decide LAMTAG app app = yes refl
decide LAMTAG app λ   = no λ ()
decide LAMTAG λ   app = no λ ()
decide LAMTAG λ   λ   = yes refl

Lam_D : One → Desc One
Lam_D ⟨⟩ = Σ_D LAMTAG λ {app → Rec_D ([] ⇒ ⟨⟩) ×_D Rec_D ([] ⇒ ⟨⟩)
                        ;  λ  → Rec_D ([] -, ([] ⇒ ⟨⟩) ⇒ ⟨⟩)}
```

Note that we do not and cannot include a tag or description for the use sites of variables in terms: use of variables in scope pertains not to the specific syntax, but to the general notion of what it is to be a syntax.

Interpreting Desc as de Bruijn Syntax. Let us give the de Bruijn interpretation of our syntax descriptions. We give meaning to Desc in the traditional manner, interpreting them as strictly positive operators in some R which gives the semantics to Rec_D. In recursive positions, the scope grows by the bindings demanded by the given Kind. At use sites, higher-kinded variables must be instantiated, just like $t[x]$ in the β-rule example: Sp_D computes the Description of the spine of actual parameters required.

```
⟦_|_⟧ : ∀ {I} → Desc I → (I → Kind I) → Kind I
⟦ Rec_D k  | R ⟧ kz = R (sort k) (kz ++ scope k)
⟦ Σ_D S T  | R ⟧ kz = Σ (Data S) λ s → ⟦ T s | R ⟧ kz
⟦ One_D    | R ⟧ kz = One
⟦ S ×_D T  | R ⟧ kz = ⟦ S | R ⟧ kz × ⟦ T | R ⟧ kz

Sp_D : Bwd (Kind I) → Desc I
Sp_D []       = One_D
Sp_D (kz -, k) = Sp_D kz ×_D Rec_D k
```

Tying the knot, we find that a term is either a variable instantiated with its spine of actual parameters, or it is a construct of the syntax for the demanded sort, with subterms in recursive positions.

```
data Tm {I} (D : I → Desc I) (i : I) kz : Set where    -- Tm D i : Kind I
  _#$_ : ∀ {jz} → (jz ⇒ i) ← kz → ⟦ Sp_D jz | Tm D ⟧ kz → Tm D i kz
  [_]  :                           ⟦ D i    | Tm D ⟧ kz → Tm D i kz
```

Interpreting Desc as co-de-Bruijn Syntax. Now let us interpret Descriptions in co-de-Bruijn style, enforcing that all variables in scope are relevant, and that binding sites expose vacuity.

```
⟦_|_⟧_R : ∀ {I} → Desc I → (I → Kind I) → Kind I
⟦ Rec_D k  | R ⟧_R = scope k ⊢ R (sort k)
⟦ Σ_D S T  | R ⟧_R = λ kz → Σ (Data S) λ s → ⟦ T s | R ⟧_R kz
⟦ One_D    | R ⟧_R = One_R
⟦ S ×_D T  | R ⟧_R = ⟦ S | R ⟧_R ×_R ⟦ T | R ⟧_R

data Tm_R {I} (D : I → Desc I) (i : I) : Kind I where
  #   : ∀ {jz} → (Va_R (jz ⇒ i) ×_R ⟦ Sp_D jz | Tm_R D ⟧_R) -:> Tm_R D i
  [_] :                            ⟦ D i    | Tm_R D ⟧_R  -:> Tm_R D i
```

We can compute co-de-Bruijn terms from de Bruijn terms, generically.

```
code  : ∀ {I} {D : I → Desc I} {i} → Tm D i       -:> (Tm_R D i ⇑_)
codes : ∀ {I} {D : I → Desc I} S  → ⟦ S | Tm D ⟧ -:> (⟦ S | Tm_R D ⟧_R ⇑_)
```

```
code                    (_#$_ {jz} x ts)  = map⇑ #   (va_R x ,_R codes (Sp_D jz) ts)
code {D = D} {i = i} [ ts ]               = map⇑ [_] (codes (D i) ts)
codes (Rec_D k)         t                 = scope k \\_R code t
codes (Σ_D S T)         (s , ts)          = map⇑ (s , _) (codes (T s) ts)
codes One_D             ⟨⟩                = ⟨⟩_R
codes (S ×_D T)         (ss , ts)         = codes S ss ,_R codes T ts
```

9 Hereditary Substitution for Co-de-Bruijn Metasyntax

Let us develop the appropriate notion of substitution for our metasyntax, *hereditary* in the sense of Watkins et al. [21]. Substituting a higher-kinded variable requires us further to substitute its parameters.

The construction of the type HSub of hereditary substitutions is subtle. We may partition the source scope into passive variables, which embed into the target, and active variables with an environment of images suited to their kinds. The HSub type is indexed by a third scope, bounding the active kinds, by way of ensuring *termination* — the oldest trick in my book [16].

```
record HSub {I} D (src trg bnd : Bwd (Kind I)) : Set where
  field pass act : Bwd (Kind I);          passive  : pass ≤ src;   active  : act ≤ src
        parti    : Cover ff passive active; passTrg : pass ≤ trg;  actBnd : act ≤ bnd
        images   : All (λ k → (scope k ⊢ Tm_R D (sort k)) ⇑ trg) act
```

Before we see how to perform a substitution, let us think how to *weaken* one: we certainly push under binders, where some $\phi : iz \leq jz$ says which iz of the jz bound variables occur in the source term. Bound variables are not substituted, so add them to the passive side, keeping the active side bounded. As with de Bruijn shifts, we must thin the images: co-de-Bruijn representation lets us thin them at the *root*!

```
wkHSub : HSub D src trg bnd → iz ≤ jz → HSub D (src ++ iz) (trg ++ jz) bnd
wkHSub {iz = iz} {jz = jz} h φ = record
  {parti = bindPassive iz; actBnd = actBnd h; passTrg = passTrg h ++≤ φ
  ; images = all (thin⇑ (oi ++≤ oe {kz = jz})) (images h)} where
  bindPassive : ∀ iz → Cover ff (passive h ++≤ oi {kz = iz}) (active h ++≤ oe {kz = iz})
```

A second handy function on HSubs whittles them down as variables are expelled from scope by thinnings in relevant pairs. We may select from an environment, but we must also refine the partition to cover just those source variables which remain, hence the selPart operation, a straightforward induction.

```
selHSub : src ≤ src′ → HSub D src′ trg bnd → HSub D src trg bnd
selHSub ψ (record {parti = c′; actBnd = θ′; images = tz′; passTrg = φ′}) =
  let ! ! ! ! φ , θ , c = selPart ψ c′ in record
    {parti = c; actBnd = θ ⨟ θ′; images = θ ≤? tz′; passTrg = φ ⨟ φ′}
selPart :  (ψ : kz ≤ kz′) → Cover ff θ′ φ′ → Σ _ λ iz → Σ _ λ jz →
  Σ (iz ≤ kz) λ θ → Σ (jz ≤ kz) λ φ → Σ (iz ≤ iz′) λ ψ0 → Σ (jz ≤ jz′) λ ψ1 → Cover ff θ φ
```

The definition of hereditary substitution is a mutual recursion, terminating because the active scope is always decreasing: hSub is the main operation on terms; hSubs and $\text{hSubs}_/$ proceed structurally, in accordance with a syntax description; hered invokes hSub hereditarily.

```
hSub  :                  HSub D src trg bnd → Tm_R D i src          → Tm_R D i ⇑ trg
hSubs  : (S : Desc I) → HSub D src trg bnd → ⟦ S | Tm_R D ⟧_R src  → ⟦ S | Tm_R D ⟧_R ⇑ trg
hSubs/ : (S : Desc I) → HSub D src trg bnd → ⟦ S | Tm_R D ⟧_R ⇑ src → ⟦ S | Tm_R D ⟧_R ⇑ trg
hered  : (jz ⊢ Tm_R D i) ⇑ trg → [] -, (jz ⇒ i) ≤ bnd → ⟦ Sp_D jz | Tm_R D ⟧_R ⇑ trg → Tm_R D i ⇑ trg
```

When hSub finds a variable, selHSub will reduce the parti to a single choice: if the variable is passive, embed it in target scope and reattach its substituted spine; if active, proceed hereditarily.

```
hSub {D = D} {i = i} h [ ts ] = map⇑ [_] (hSubs (D i) h ts)
hSub h (# {jz} (pair (only ↑ θ) ts _)) with selHSub θ h | hSubs/ (Sp_D jz) h ts
... | record {parti = _css {both = ()} _}                          | ts'
... | record {parti = czz cs'; passTrg = φ}                        | ts' = map⇑ # (va_R φ ,_R ts')
... | record {parti = czz c's; actBnd = θ'; images = [] -, im} | ts' = hered im θ' ts'
```

To substitute a variable hereditarily, find it in the bound: the scope of its kind becomes the new, *structurally smaller* bound. Helper function part partitions passive free variables from active bound variables, while spAll converts the spine to an environment of images.

```
hered                    im          (θ' o') ts' = hered im θ' ts'
hered {D = D} {trg = trg} ((φ \ t) ↑ ψ) (θ' os) ts' = let ! ! c = part _ _ in
  hSub (record {parti = c; actBnd = φ; images = φ ≤? spAll ts'; passTrg = ψ}) t where
    spAll : ∀ {kz} → ⟦ Sp_D kz | Tm_R D ⟧_R ⇑ trg → All _ kz
    part  : ∀ kz iz → Σ (kz ≤ (kz ++ iz)) λ θ → Σ (iz ≤ (kz ++ iz)) λ θ' → Cover ff θ θ'
```

In the structural part of the algorithm, we may exploit our richer usage information to stop as soon as the active variables have all left scope, thinning the remaining passive variables with no further traversal. The lemma allLeft shows that if the right of a partition is empty, the left must be full.

```
hSubs (Rec_D k) h (φ \ t)      = scope k \\_R hSub (wkHSub h φ) t
hSubs (Σ_D S T) h (s , ts)     = map⇑ (s ,_) (hSubs (T s) h ts)
hSubs One_D     h ⟨⟩            = ⟨⟩_R
hSubs (S ×_D T) h (pair s t _) = hSubs/ S h s ,_R hSubs/ T h t
hSubs/ S h' (ts ↑ θ) with selHSub θ h'
hSubs/ S h' (ts ↑ θ) | record {parti = c; images = []; passTrg = φ} rewrite allLeft c = ts ↑ φ
hSubs/ S h' (ts ↑ θ) | h = hSubs S h ts
```

10 Discussion

We have a universe of syntaxes with metavariables and binding, where the Description of a syntax is interpreted as the co-de-Bruijn terms, ensuring intrinsically that unused variables are discarded not at the *latest* opportunity (as in de Bruijn terms), nor at an *arbitrary* opportunity (as in one of Bird and Paterson's variants [7], or with Hendriks and van Oostrom's 'adbmal' operator [13], both of which reduce the labour of shifting at the cost of nontrivial α-equivalence), but at the *earliest* opportunity. Hereditary substitution exploits usage information to stop when there is nothing to substitute, shifts without traversal, and is, moreover, structurally recursive on the *active scope*.

Recalling Atkey's Battlestar-Galactica-inspired quip about de Bruijn indices being a 'Cylon detector', co-de-Bruijn representation is even less suited to human comprehension, but its informative precision makes it all the more useful for machines. Dependency checking is direct, so syntactic forms like vacuous functions or η-redexes are easy to spot.

Co-de-Bruijn representation could lead to more efficient implementations of normalization and of metavariable instantiation. The technique may be readily combined with representing terms as trees whose top-level leaves are variable uses and top-level nodes are just those (now easily detected) where paths to variables split: edges in the tree are *closed* one-hole contexts, jumped over in constant time [9].

I see two high-level directions emerging from this work. Firstly, the generic treatment of syntax with *meta*variables opens the way to the generic treatment of *metatheory*. Even without moving from scope-safe to type-safe term representations, we can generate the inductive relations we use to define notions such as reduction and type synthesis in a universe, then seek to capture good properties (e.g., stability under substitution, leading to type soundness) by construction. Co-de-Bruijn representations make it easy to capture properties such as variable non-occurrence in the syntax of formulæ, and might also serve as the target term representation for algorithms extracted generically from the rules.

Secondly, more broadly, this work gives further evidence for a way of programming with strong invariants and redundant but convenient information caches without fear of bugs arising from inconsistency. We should put the programmer in charge! Dependent types should let us take control of data representations and optimise them to support key operations, but with the invariants clearly expressed in code and actively supporting program synthesis.

Only a fool would attempt to enforce the co-de-Bruijn invariants without support from a typechecker, so naturally I have done so: using Haskell's `Integer` for bit vectors (making `-1` the identity of the unscoped thinning *monoid*), I implemented a dependent type system, just for fun. It was Hell's delight, even with the Agda version to follow. I was sixteen again.

Acknowledgements. EPSRC project EP/M016951/1 *Homotopy Type Theory: Programming and Verification* funded this work. My Mathematically Structured Programming colleagues at Strathclyde made me get these ideas in shape: Fredrik Nordvall Forsberg offered particularly useful advice about what to omit. Philippa Cowderoy's use of *information effects* for typing contexts increased my sensitivity to the signposting of discards and duplications. An EU TYPES Short Term Scientific Mission brought Andrea Vezzosi to Strathclyde, provoking ideas and action for further work. Invitations to present at *Trends in Functional Programming 2017* (Canterbury) and to my old Nottingham friends (notably Thorsten Altenkirch) helped me find the words. James McKinna and Randy Pollack remain an inspiration. And thanks, tweeps!

References

[1] Andreas Abel (2006): *Implementing a Normalizer Using Sized Heterogeneous Types*. In Conor McBride & Tarmo Uustalu, editors: *Workshop on Mathematically Structured Functional Programming, MSFP@MPC 2006, Kuressaare, Estonia, July 2, 2006.*, Workshops in Computing, BCS.

[2] Guillaume Allais, James Chapman, Conor McBride & James McKinna (2017): *Type-and-scope safe programs and their proofs*. In Yves Bertot & Viktor Vafeiadis, editors: *Proceedings of the 6th ACM SIGPLAN Conference on Certified Programs and Proofs, CPP 2017, Paris, France, January 16-17, 2017*, ACM, pp. 195–207, doi:10.1145/3018610.3018613. Available at `http://doi.acm.org/10.1145/3018610.3018613`.

[3] Thorsten Altenkirch, Martin Hofmann & Thomas Streicher (1995): *Categorical Reconstruction of a Reduction Free Normalization Proof*. In David H. Pitt, David E. Rydeheard & Peter T. Johnstone, editors: *Category*

Theory and Computer Science, 6th International Conference, CTCS '95, Cambridge, UK, August 7-11, 1995, Proceedings, *Lecture Notes in Computer Science* 953, Springer, pp. 182–199.

[4] Thorsten Altenkirch & Bernhard Reus (1999): *Monadic presentations of lambda-terms using generalized inductive types*. In: *Computer Science Logic 1999*.

[5] Francoise Bellegarde & James Hook (1995): *Substitution: A formal methods case study using monads and transformations*. *Science of Computer Programming*.

[6] Nick Benton, Chung-Kil Hur, Andrew Kennedy & Conor McBride (2012): *Strongly Typed Term Representations in Coq*. *J. Autom. Reasoning* 49(2), pp. 141–159.

[7] Richard Bird & Ross Paterson (1999): *de Bruijn notation as a nested datatype*. *Journal of Functional Programming* 9(1), pp. 77–92.

[8] Nicolas G. de Bruijn (1972): *Lambda Calculus notation with nameless dummies: a tool for automatic formula manipulation*. *Indagationes Mathematicæ* 34, pp. 381–392.

[9] Lucas Dixon, Peter Hancock & Conor McBride (2007): *Why walk when you can take the tube?* Available at `http://strictlypositive.org/Holes.pdf`. Unpublished draft.

[10] Jeremy Gibbons (2017): *APLicative Programming with Naperian Functors*. In Hongseok Yang, editor: *Programming Languages and Systems - 26th European Symposium on Programming, ESOP 2017, Held as Part of the European Joint Conferences on Theory and Practice of Software, ETAPS 2017, Uppsala, Sweden, April 22-29, 2017, Proceedings*, *Lecture Notes in Computer Science* 10201, Springer, pp. 556–583.

[11] Healfdene Goguen & James McKinna (1997): *Candidates for Substitution*. Technical Report ECS-LFCS-97-358, University of Edinburgh.

[12] Robert Harper, Furio Honsell & Gordon D. Plotkin (1993): *A Framework for Defining Logics*. *J. ACM* 40(1), pp. 143–184.

[13] Dimitri Hendriks & Vincent van Oostrom (2003): *adbmal*. In Franz Baader, editor: *Automated Deduction - CADE-19, 19th International Conference on Automated Deduction Miami Beach, FL, USA, July 28 - August 2, 2003, Proceedings*, *Lecture Notes in Computer Science* 2741, Springer, pp. 136–150.

[14] Chantal Keller & Thorsten Altenkirch (2010): *Hereditary Substitutions for Simple Types, Formalized*. In Venanzio Capretta & James Chapman, editors: *Proceedings of the 3rd ACM SIGPLAN Workshop on Mathematically Structured Functional Programming, MSFP@ICFP 2010, Baltimore, MD, USA, September 25, 2010.*, ACM, pp. 3–10.

[15] Conor McBride (2000): *Dependently typed functional programs and their proofs*. Ph.D. thesis, University of Edinburgh, UK.

[16] Conor McBride (2003): *First-order unification by structural recursion*. *J. Funct. Program.* 13(6), pp. 1061–1075.

[17] Spike Milligan (1972): *The Last Goon Show of All*. BBC Radio 4.

[18] Ulf Norell (2008): *Dependently Typed Programming in Agda*. In Pieter W. M. Koopman, Rinus Plasmeijer & S. Doaitse Swierstra, editors: *Advanced Functional Programming, 6th International School, AFP 2008, Heijen, The Netherlands, May 2008, Revised Lectures*, *Lecture Notes in Computer Science* 5832, Springer, pp. 230–266.

[19] Masahiko Sato, Randy Pollack, Helmut Schwichtenberg & Takafumi Sakurai (2013): *Viewing λ-terms through Maps*. *Indagationes Mathematicæ* 24(4).

[20] Philip Wadler (1989): *Theorems for Free!* In: *Proceedings of the Fourth International Conference on Functional Programming Languages and Computer Architecture*, FPCA '89, ACM, New York, NY, USA, pp. 347–359.

[21] Kevin Watkins, Iliano Cervesato, Frank Pfenning & David Walker (2003): *A Concurrent Logical Framework: The Propositional Fragment*. In Stefano Berardi, Mario Coppo & Ferruccio Damiani, editors: *Types for Proofs and Programs, International Workshop, TYPES 2003, Torino, Italy, April 30 - May 4, 2003, Revised Selected Papers*, *Lecture Notes in Computer Science* 3085, Springer, pp. 355–377.